
Im folgenden findet sich eine Anleitung zum Einlesen der BesT-Daten aus den Beispieldatei-Vorlagen sowie die 

Umkodierung und Weiterverarbeitung mit Summenwerten von Subskalen.  

Dabei werden die Daten aus den jeweiligen Excel-files eingelesen und in SAS work files gespeichert (Beispiel: 

„bestk“ für die Kinderbögen). Das Einlesen kann selbstverständlich auch aus anderen Datenformaten erfolgen, 

ebenso kann die work file-Bezeichnung geändert werden, muss aber dann jeweils angepasst werden. 

Wichtig ist das Umpolen der dafür vorgesehenen Items im Jugendlichen- und im Elternbogen. Die Auswahl der 

Items für die Bildung der Summenwerte für Gesamt- und Subskalen wird ebenfalls dargestellt. Beides kann 

natürlich auch mit anderen Statistikprogrammen erledigt werden, wenn ein anderes bevorzugt wird. Die 

vorliegende SAS-Syntax soll vor allem den Algorithmus darstellen und weitere Skripte sind geplant (z.B. in R). 

Rückmeldungen sind willkommen, ebenso eventuelle Skripte für andere Statistikpakete! 

*-- einlesen Best-X Daten; 

proc import datafile="c:\sas_mein\best-kj\BesT_K_Klinikname_2021.xlsx" out=bestk 

        dbms=xlsx replace; getnames=yes;  

proc import datafile="c:\sas_mein\best-kj\BesT_J_Klinikname_2021.xlsx" out=bestj 

        dbms=xlsx replace; getnames=yes;  

proc import datafile="c:\sas_mein\best-kj\BesT_E_Klinikname_2021.xlsx" out=beste 

        dbms=xlsx replace; getnames=yes;  

run; 

 

*--- einige initiale Auswertungen und Checks (am Beispiel Kinderbogen); 

 

proc contents data=bestk; * zeige Variablen im file und Eigenschaften; 

  run; 

proc freq data=bestk;  

  table  geschlecht alter station; * Häufigkeiten; 

run; 

proc means data=bestk;  

  var best_nK1 -- best_nK25; * N, Mittelwerte, SD, Min, Max;  

run; 

proc freq data=bestk; * Häufigkeiten ausgewählter Variablen; 

  table best_nK1 best_nK18 best_nK25 best_nK26a best_nK26b ;  

run; 

 

*----- Kinderbögen bearbeiten ---; 

data ki; set bestk; 

   if auffaellig_nk = 'x' then delete; * auffällige Bögen entfernen; 

   if nummer ne ' '; * filtert leere Datensätze weg;  

   array bestk{*} best_nK1 -- best_nK20; 

   if best_nK18 >= 0 then tagesklinik = 0; * was angekreuzt?; 

   if best_nK18 = 7 then do;  

        tagesklinik = 1; * war in Tagesklinik; 

        best_nK18 = 0;   * wird zu missing im nächsten DO-loop; 

      end; 

   K24_keineZwangsmass = .; 

   if best_nK24 = 0 then K24_keineZwangsmass = 1; * keine Zwangsmass. angekreuzt; 

   K25_keinAnfassen = .; 

   if best_nK25 = 0 then K25_keinAnfassen = 1; * kein Anfassen angekreuzt; 

   * umpolen Kinderversion; 

   do i=1 to 20; 

      if bestk[i] = 0 then bestk[i] = .; * nicht stattgefunden als missing; 

   end; 

   do i=1 to 20; 

      bestk[i] = 6 - bestk[i]; * umpolen alle Items; 

   end; 

   best_nk19 = 6 - best_nk19; * wieder zurück-umpolen (Angstitem); 

   * umpolen der beiden Items 24 und 25; 

   if best_nk24 = 0 then best_nk24 = .; * nicht stattgefunden als missing; 

   if best_nk25 = 0 then best_nk25 = .; * nicht stattgefunden als missing; 

   best_nk24 = 6 - best_nk24; * umpolen; 

   best_nk25 = 6 - best_nk25; * umpolen; 

   label 

            best_nk1 = "Zeit gefallen" 

  best_nk2 = "mit Aufenthalt zufrieden" 

  best_nk3 = "Problem jetzt weg" 

  best_nk4 = "Zimmer gefallen" 

  best_nk5 = "Station gefallen" 

  best_nk6 = "Essen geschmeckt" 

  best_nk7 = "bei Untersuchung wohl gefühlt" 

  best_nk8 = "Aufklärung Medikamente" 

  best_nk9 = "Therapeut verstanden" 

  best_nk10= "Therapeut alles erzählen" 

  best_nk11= "Therapeut Zeit gehabt" 

  best_nk12= "Betreuer Problem verstanden" 

  best_nk13= "Betreuer nett" 

  best_nk14= "Personal ansprechbar" 

  best_nk15= "angeklopft an Zimmer" 

  best_nk16= "mit anderen verstanden" 

  best_nk17= "mitentscheiden Behandlung" 

  best_nk18= "Ausgangsregelung" 

  best_nk19= "Angst, dass andere mitkriegen" 

  best_nk20= "wieder in unsere Klinik" 

  best_nk24= "Zwangsmassnahmen erklärt" 

  best_nk25= "zu nahe oder angefasst" 

   ; 

  * Summenscores Kinder; 

  md_best_kneu = nmiss(of best_nk1 - best_nk18 best_nk20 best_nk24); * Anzahl missings; 

  genfak_K     = mean( of best_nk1 - best_nk18 best_nk20 best_nk24);   

  fak1_K_ThBez = mean(of  best_nk8, best_nk9, best_nk10, best_nk11, best_nk12,  

                          best_nk13, best_nk14, best_nk15, best_nk17, best_nk18, best_nk24); 

  fak2_K_Umgeb = mean(of  best_nk1, best_nk2, best_nk4, best_nk5, best_nk6, best_nk20); 

  label 

     genfak_K      = "Gesamtwert (Kinder)" 

 fak1_K_ThBez  = "Therapeutische Beziehung" 

 fak2_K_Umgeb  = "Umgebung" 

 md_best_kneu  = "Anzahl missings (Kinder)" 

 ; 

run; 

 

*--- check neue Variablen; 

* Häufigkeiten der Kinder mit x missings; 

proc freq data=ki;  

   table  md_best_kneu;  

run; 

* check deskriptive Maße und Minimum und Maximum; 

proc means data=ki;  

   var best_nk1 - best_nK25 genfak_K fak1_K_ThBez fak2_K_Umgeb; 

run; 

 

 

*----- Elternboegen bearbeiten ---; 

data el; set bestk bestj beste;  * hole Elternboegen aus allen 3 Dateien; 

   if auffaellig_ne = 'x' then delete; * auffaellige Bögen weg; 

   if nummer ne ' '; * filtert leere Datensätze;  

   array beste{*} best_ne1 -- best_ne23; 

   * umpolen Elternversion; 

   do i=1 to 23; 

      if beste[i] = 0 then beste[i] = .; * nicht stattgefunden als missing; 

   end; 

   do i=4 to 21; 

      beste[i] = 6 - beste[i]; * umpolen Item 4 - 21; 

   end; 

  label 

               best_ne1 = "insgesamt zufrieden" 

  best_ne2 = "Aufenthalt Kind geholfen" 

  best_ne3 = "Aufenthalt uns geholfen" 

  best_ne4 = "Aufklärung Krankheit" 

  best_ne5 = "Aufklärung Medikamente" 

  best_ne6 = "ernst nehmen (Therapeut)" 

  best_ne7 = "Einzelgespräche wirksam" 

  best_ne8 = "Familiengespräche wirksam" 

  best_ne9 = "Aufenthalt bringt viel" 

  best_ne10= "Mitsprache bei Therapien" 

  best_ne11= "Schweigepflicht" 

  best_ne12= "Zwangsmassnahmen" 



  best_ne13= "Mitsprache Kind bei Zimmer" 

  best_ne14= "Wochenendbeurlaubung" 

  best_ne15= "sanitäre Anlagen" 

  best_ne16= "Station schön" 

  best_ne17= "Mitsprache Entlassung" 

  best_ne18= "mehr Einzelgespräche" 

  best_ne19= "Privatsphäre Kind" 

  best_ne20= "Kind zurückziehen auf Stat." 

  best_ne21= "Behandlungsziele absprechen" 

  best_ne22= "wieder hierher" 

  best_ne23= "Angst, dass andere erfahren" 

   ; 

  * Summenscores Eltern; 

  md_best_eneu = nmiss( of best_ne1 - best_ne22); * Anzahl missings (ohne das Stigma-Item); 

  genfak_E = mean( of best_ne1 - best_ne22);  

  fak1_E_ThBez = mean(of  best_ne4,  best_ne5,  best_ne6,  best_ne7,  best_ne8, best_ne9,  

                          best_ne10, best_ne11, best_ne17, best_ne18, best_ne21); 

  fak2_E_Umgeb = mean(of  best_ne13, best_ne14, best_ne15, best_ne16, best_ne19, best_ne20); 

  fak3_E_AllgZuf = mean(of  best_ne1, best_ne2, best_ne3, best_ne22); 

  label 

     genfak_E       = "Gesamtwert (Eltern)" 

     fak1_E_ThBez   = "Beziehung zu Therapeut_in" 

     fak2_E_Umgeb   = "Umgebung" 

     fak3_E_AllgZuf = "Allgemeine Zufriedenheit" 

     md_best_kneu   = "Anzahl missings (Eltern)" 

  ; 

run; 

 

*-- check neue Variablen; 

* Häufigkeiten der Elternbögen mit missings; 

proc freq data=el;  

  table md_best_eneu; 

run; 

* deskriptive Maße und Min und Max; 

proc means data=el;  

  var best_ne1 - best_ne23 genfak_E -- fak3_E_AllgZuf;  

run; 

 

 

*----- Jugendlichenboegen ---; 

data ju; set bestj; 

   if auffaellig_nj = 'x' then delete; * auffällige Bögen weg; 

   if nummer ne ' '; * entfernt leere Datensätze;  

   array bestj{*} best_nj1 -- best_nj29; 

   J10_keineSchule = .; 

   if best_nj10 = 0 then J10_keineSchule = 1; * kein Schulbesuch angekreuzt; 

   * umpolen Jugendlichenversion; 

   do i=1 to 29; 

      if bestj[i] = 0 then bestj[i] = .; * nicht stattgefunden als missing; 

   end; 

   do i=4 to 25; 

      bestj[i] = 6 - bestj[i]; * umpolen Item 4 - 25; 

   end; 

   label 

               best_nj1 = "insgesamt zufrieden" 

  best_nj2 = "Aufenthalt geholfen" 

  best_nj3 = "motiviert, mitzuarbeiten" 

  best_nj4 = "Aufklärung Krankheit" 

  best_nj5 = "Aufklärung Medikamente" 

  best_nj6 = "ernst nehmen (Therapeut)" 

  best_nj7 = "Einzelgespräche wirksam" 

  best_nj8 = "Familiengespräche wirksam" 

  best_nj9 = "ernst nehmen (Betreuer)" 

  best_nj10= "ernst nehmen (Lehrer)" 

  best_nj11= "Schweigepflicht" 

  best_nj12= "Zwangsmassnahmen" 

  best_nj13= "Ausgangsregelung" 

  best_nj14= "Essen" 

  best_nj15= "sanitäre Anlagen" 

  best_nj16= "sich zurückziehen mit Eltern" 

  best_nj17= "Station schön" 

  best_nj18= "Schulangebot" 

  best_nj19= "Mitsprache Therapien" 

  best_nj20= "Mitsprache Zimmer" 

  best_nj21= "Aufklärung poststationär" 

  best_nj22= "Klima Jugendliche" 

  best_nj23= "Privatsphäre" 

  best_nj24= "allein sein können" 



  best_nj25= "Ziele absprechen" 

  best_nj26= "Erwartungen erfüllt" 

  best_nj27= "wieder hierher" 

  best_nj28= "Angst, dass andere erfahren" 

  best_nj29= "zu nahe oder angefasst" 

 ; 

  * Summenscores Jugendliche; 

  md_best_jneu = nmiss( of best_nj1 - best_nj27); * missings (ohne Stigma und anfassen); 

  genfak_J = mean( of best_nj1 - best_nj27);  

  fak1_J_ThBez = mean(of  best_nj4, best_nj5,  best_nj6,  best_nj7,  best_nj8,  

                          best_nj9, best_nj10, best_nj11, best_nj21, best_nj25); 

  fak2_J_Umgeb = mean(of  best_nj13, best_nj14, best_nj15, best_nj16, best_nj17,  

                          best_nj20, best_nj22, best_nj23, best_nj24); 

  fak3_J_AllgZuf = mean(of  best_nj1, best_nj2, best_nj3, best_nj26, best_nj27); 

  label 

     genfak_J       = "Gesamtwert (Jugendliche)" 

     fak1_J_ThBez   = "Therapeutische Beziehung" 

     fak2_J_Umgeb   = "Umgebung" 

     fak3_J_AllgZuf = "Allgemeine Zufriedenheit" 

     md_best_jneu   = "Anzahl missings (Jugendliche)" 

  ; 

run; 

 

*--- check neue Variablen; 

proc freq data=ju;  

  * table  erhebungsjahr; *best_nj1 best_nj5 best_nj10 best_nj25 best_nj28 

          J5_keineMedi J10_keineSchule station;  

  table station J10_keineSchule md_best_jneu; 

run; 

proc means data=ju;  

   var    best_nj1 - best_nj29 genfak_J -- fak3_J_AllgZuf;  

run; 

 

*----- Start automatischer Output mit ODS; 

* (geht natürlich auch ohne ODS, mit output auf Bildschirm); 

ods rtf file='c:\sas_mein\best-kj\Erg_best-kijuel-Klinikname_2021.rtf'; 

ods NOPTITLE; 

  

*  Kinder; 

data raus; set ki; 

title 'Haeufigkeitsverteilung (nach vorhandenen Kinderbogen)'; 

proc freq; 

  table geschlecht alter station verweil_14 beschluss; 

  where best_nk1 > 0; * Filterkriterium: Item 1 beantwortet; 

run; 

title 'Gesamtauswertung Kinderbogen'; 

proc means fw=4;  

  var best_nk1 - best_nk20 best_nk24 best_nk25  

      genfak_K fak1_K_ThBez fak2_K_Umgeb;  

  run; 

title 'Unterteilung nach Geschlecht (Kinder)'; 

proc sort data=raus; by geschlecht; run; 

proc means fw=4; 

  var best_nk1 - best_nk20 best_nk24 best_nk25 

      genfak_K fak1_K_ThBez fak2_K_Umgeb;  

  by geschlecht; 

  run; 

 

*-- usw., analog für Jugendliche und Elternbögen.....; 

 

ods rtf close; 

 


